

IBM

@server

iSeries

システム遮断制御の計画







@server

iSeries

システム遮断制御の計画

© Copyright International Business Machines Corporation 1998, 2002. All rights reserved.

© Copyright IBM Japan 2002

---

# 目次

---

第 1 部 電源管理プログラムを使用したサーバー遮断の制御 . . . . .	1
第 1 章 コードに関する特記事項 . . . . .	3
第 2 章 電源管理プログラムを使用可能にし、電源障害時のシステム活動を制御する . . . . .	5
第 3 章 電源管理プログラムの設定 . . . . .	7
第 4 章 電源管理プログラムの作成 . . . . .	9
第 5 章 例: 電源管理プログラム . . . . .	11
例: 電源管理 CL プログラム . . . . .	11
例: 電源管理 CL プログラムのテスト . . . . .	14
第 6 章 制御遮断の概念 . . . . .	17
バッテリー・バックアップ装置 . . . . .	17
電源管理プログラムが存在しない場合の無停電電源装置条件の処理 . . . . .	17
無停電電源装置の IPL 考慮事項 . . . . .	17
停電時の制御遮断 . . . . .	18
電源回復 IPL システム値 (QPWRRSTIPL) . . . . .	18
無停電電源装置 . . . . .	19
無停電電源装置遅延時間システム値 (QUPSDLYTIM) . . . . .	19
QUPSDLYTIM 値およびアクション・フローチャートに関する解説 . . . . .	21
無停電電源装置メッセージ待ち行列システム値 (QUPSMMSGQ) . . . . .	22
無停電電源装置メッセージ . . . . .	22
無停電電源装置からのバッテリー低下状態信号 . . . . .	24
QUPSDLYTIM 機能の時間軸に関する図の解説 . . . . .	25



---

## 第 1 部 電源管理プログラムを使用したサーバー遮断の制御

電源管理プログラムを電源保護装置と共に使用すると、停電中のシステムの中断を最小限にとどめることができるようになります。電源保護 (たとえば UPS) 装置を使用すると、一時的に通常電源が中断しても、システムに電力の供給が行われます。このようにして、システム機能が異常終了することを避けることができます。制御遮断機構により、システムの電源遮断が可能な限り円滑に行われ、再 IPL 時に不都合な影響を最小限にとどめます。

作業に取りかかるためのステップを次に示します。

- **電源管理プログラムの使用可能化**

電源管理プログラムを使用可能にし、電源障害時のシステム活動を制御する。これは、電源遮断をできるだけ円滑に行うための 1 つの方法です。

- **電源管理プログラムの設定**

iSeries サーバーでの電源管理プログラムの設定の例。電源管理プログラムを使用するために必要な手順をステップごとに示します。

- **電源管理プログラムの作成**

電源管理プログラムを作成すると、何ら特別な処理なしに、短時間の電源障害に対処できるようになります。また、電源がすぐに回復しない場合でも、通常の電源遮断が行えるようになります。

特定のシステム要件に応じて、ここに挙げられている CL プログラムの例を調整することができます。また、ユーザーが作成した電源管理プログラムをテストすることもできます。

制御遮断の概念と定義については、17 ページの『第 6 章 制御遮断の概念』を参照してください。

注: 重要な法的情報については、3 ページの『第 1 章 コードに関する特記事項』をお読みください。





---

## 第 1 章 コードに関する特記事項

本書には、プログラミングのサンプルが含まれています。

IBM® は、お客様に、すべてのプログラミング・コードをサンプルとして使用することができる非独占的な著作権のあるライセンスを許諾します。お客様は、このサンプル・コードから、お客様特有のニーズに合わせた類似のプログラムを作成することができます。

すべてのサンプル・コードは、例として示す目的でのみ、IBM により提供されます。このサンプル・プログラムは、あらゆる条件下における完全なテストを経ていません。従って IBM は、これらのプログラムについて信頼性、利便性もしくは機能性について、問題がないことをほのめかしたり、保証することはできません。

ここに含まれるすべてのプログラムは、現存するままの状態を提供され、いかなる保証条件も適用されません。不侵害、商品性、特定目的適合性に関する黙示の保証の適用も一切ありません。



---

## 第 2 章 電源管理プログラムを使用可能にし、電源障害時のシステム活動を制御する

システム・ソフトウェア・サポートは、バッテリー機構用、または無停電電源装置用でも、基本的には同じものです。環境によっては、無停電電源装置がシステムへの電源供給を開始した時、あるいは電力が不安定な時に、異なった処置を行わせたい場合があります。そのような状況に対し、電源管理プログラムは次のいずれかの方法で対処できます。

- 特定のメッセージを対話式ユーザーに送る
- 電源遮断に備えて、バッチ・ジョブとサブシステムを終了する
- 無停電電源装置の処理を制御するシステム値を動的に変更する
- PWRDWN SYS コマンドを発行してシステムの電源を遮断する

電源管理プログラムを使用していることを指定するには、QUPSMMSGQ システム値を、ユーザーが作成した待ち行列の名前に変更します。システムは、QSYSOPR と指定した待ち行列の両方に、同一のメッセージを送信するようになります。QUPSDLYTIM システム値を \*NOMAX に変更してください。

メッセージ待ち行列の処理に使用するプログラムは活動状態でなければなりませんし、待ち行列を割り振る必要もあります。プログラムが、QUPSMMSGQ に指定されている待ち行列を割り振っていない場合、システムは電源管理プログラムが存在しないものと見なします。



## 第 3 章 電源管理プログラムの設定

次に、完全無停電電源装置を接続した場合に、iSeries サーバーで電源管理プログラムの設定を行うための例を示します。この例は、QCTL が制御サブシステムであることを前提としています。完全無停電電源装置を使用するときのサンプル・プログラムについては、11 ページの『例: 電源管理 CL プログラム』を参照してください。

1. 電源管理プログラムは、その性質上、重要な場面で使用することになるため、電源管理プログラムが使用するオブジェクトを専用のライブラリーに分離して、他のユーザーから保護しなければなりません。これは以下のようにして行います。

```
CRTLIB LIB(UPSLIB) AUT(*EXCLUDE) CRTAUT(*EXCLUDE)
```

2. 電源管理プログラムには、専用のメッセージ待ち行列が必要です。このため、固有のメッセージ待ち行列を作成し、それを他のユーザーおよび一般システムが使用しないようにしなければなりません。これは以下のようにして行います。

```
CRTMSGQ MSGQ(UPSLIB/UPSMGQ) AUT(*EXCLUDE)
```

3. CL 電源管理プログラムを作成し、これを他のユーザーが使用できないようにします。これは以下のようにして行います。

```
CRTCLPGM PGM(UPSLIB/UPSPGM) AUT(*EXCLUDE)
```

4. 制御サブシステムの始動時に自動的に開始する、電源管理プログラムのジョブ記述を作成します。

```
CRTJOB JOB(UPSLIB/UPSJOB) JOBQ(QSYS/QCTL2)  
JOBPTY(1) RQSDTA('CALL UPSLIB/UPSPGM')  
AUT(*EXCLUDE) USER(XXXX)
```

注: ジョブ記述を自動開始ジョブとして使用するには、ユーザー・プロファイルを提供しなければなりません。

5. 現在の制御サブシステム記述をコピーし、代替制御サブシステム記述を作成します。これは以下のようにして行います。

```
CRTDUPOBJ OBJ(QCTL) FROMLIB(QSYS)  
OBJTYPE(*SBSD) TOLIB(QSYS) NEWOBJ(QCTL2)
```

6. すべてのサブシステムを開始するようにスタートアップ・プログラムを変更します。システム値 QCTLSBSD が QCTL2 と等しいかどうかを確認するための検査を組み込む必要があります。名前およびライブラリーについては、システム値 QSTRUPPGM を参照してください。スタートアップ・プログラムを変更しない場合には、QSYS または QGPL 内の QCTL2 は検査されず、スタートアップ・プログラムはサブシステムの残りを開始しないで終了します。

7. 自動開始ジョブ項目を代替制御サブシステム記述に追加します。これは以下のようにして行います。

```
ADDAJE SBS(QSYS/QCTL2) JOB(QSYS/QCTL2)  
JOB(UPSLIB/UPSJOB)
```

8. 代替制御サブシステムを使用するために、制御サブシステム・システム値を変更します。以下のようにして行います。

```
CHGSYSVAL SYSVAL(QCTLSBSD) VALUE('QCTL2')
```

9. プログラムが電源異常を処理できるように、システム値を変更します。以下のようにして行います。

```
CHGSYSVAL SYSVAL(QUPSMGQ) VALUE('UPSMGQ UPSLIB')  
CHGSYSVAL SYSVAL(QUPSDLYTIM) VALUE(*NOMAX)
```

10. システムの IPL を実行し、新しい制御サブシステム記述を有効にします。以下のようにして行います。

```
PWRDWSYS OPTION(*IMMED) RESTART(*YES)
```



## 第 4 章 電源管理プログラムの作成

電源管理プログラムは、IPL のたびに活動化され、活動状態のままでなければなりません。電源管理プログラムは、実行管理機能サブシステムの仕様が許す活動レベルになければなりません。

QUPSMGQ で指定したメッセージ待ち行列は、無停電電源装置メッセージ処理で使用されます。通常、以下のコマンドを指定すると、プログラムは待ち行列を割り振ります。

```
ALCOBJ OBJ(xxx/yyy *MSGQ *EXCL)
```

着信するメッセージのうち、処理を必要とする重要なメッセージを以下に挙げます。

- **CPF1816: &1 にシステムの通常電源の障害が起こった。** (このメッセージは、バッテリー機構と全機能電源機構に適用されます)
- **CPF1817: &1 にシステムの通常電源が復元された。** (このメッセージは、バッテリー機構と全機能電源機構に適用されます)
- **CPI0994: システムの電源が復元された。** (このメッセージは、機能制限のある無停電電源装置に適用されます)
- **CPI0963: システムは補助電源になっている。** (このメッセージは、IPL 中にシステム電源障害が起きた場合に適用されます)

他のメッセージは無視しても構いません。

短時間の電源障害であれば、プログラムは特別な処理なしに対処できます。たとえば、CPF1816 メッセージが着信した場合、このメッセージが発生したことを示すスイッチをプログラムの中に設定できます。次いで、プログラムは RCVMSG に WAIT(10) を指定して実行し、10 秒でタイムアウトになるようにできます。タイムアウトになる前に CPF1817 メッセージを受信した場合は、スイッチをリセットして他の処置を実行できます。

電源がすぐに回復しない場合でも通常の電源遮断が行えるよう、プログラムを作成することができます。たとえばリモート・ワークステーションがあり、これが依然として活動状態である場合、それらのワークステーションに対して、直ちにサインオフするよう要求するメッセージを送信することができます。ENDSBS OPTION(\*CNTRLD) を出して、新しいワークステーションがサインオンしたり、新しいバッチ作業が始まったりすることを避けることができます。バッチ・ジョブを実行している場合、以下のコマンドを使用してそれらを終了することができます。

```
ENDJOB OPTION(*CNTRLD)
```

これにより、標識はジョブ終了に設定されます。いくつかの高水準言語および制御言語では、制御を受けた ENDJOB が指定済みかどうかをプログラム内部でテストすることが可能です。プログラムが自分で終了しない場合は、デフォルトとして ENDJOB (30 秒) が使用されます。

RCVMSG WAIT(120) などの 2 番目のタイマーをプログラムに設定することもできます。通常電源が回復しなかった場合に、PWRDWN SYS OPTION(\*IMMED) コマンドを発行することができます。待ち時間は、バッテリー時間と電源遮断に必要な時間に基づいて指定しなければなりません。

QUPSMGQ システム値にメッセージ待ち行列 を指定し、QUPSDLYTIM に \*NOMAX を指定する場合は、次の条件が適用されます。

- 指定したメッセージ待ち行列は、CPF1816 メッセージが生じたときにプログラムによって割り振られなければなりません。

- 指定するメッセージ待ち行列がワークステーション・メッセージ待ち行列である場合、中断モードまたは通知モードになければなりません。

上記の条件が満たされないと、システムは電源管理プログラムが存在しないと見なし、システムは電源遮断されます。

**注:** システムが制限状態になると (たとえば ENDSBS \*ALL)、無停電電源装置を処理するプログラムは活動状態ではなくなります。このため、無停電電源装置を処理するための別の方法、およびシステムが制限状態にあるときに起こり得る電源障害に対処する方法を用意しておく必要があります。

たとえば、SAVSYS (システムの保管) または RCLSTG (記憶域の再利用) を実行しているときに、すべてのサブシステムが終了すると、無停電電源装置プログラムは活動状態ではなくなります。単一のワークステーション・ジョブだけが活動状態になります。別の方法として、次の処置のいずれかを実行できます。

1. すべてのサブシステムが終了したら、コマンド行を使って、システム値 QUPSMGQ に指定されているメッセージ待ち行列のモードを \*BREAK に変更します。これにより、すべての無停電電源装置メッセージが、中断メッセージとして、ワークステーションにサインオンしているユーザーに送信されます。この方法をとった場合、電源障害が起こったときに、ユーザー自身で処置を決定することになります。
2. システム値 QUPSDLYTIM を \*NOMAX 以外の値に変更します。(たとえば、無停電電源装置が電力供給できる、分単位の時間)。この方式により、システムが即時に電源遮断するのを防ぐことができます。電源障害が生じ、その時間がシステム値 QUPSDLYTIM に指定した時間を超えた場合は、即時電源遮断が生じます。
3. 無停電電源装置を処理するプログラムが、BREAK HANDLING プログラムを使用できるように修正します。このプログラムは、システムが制限状態にあるときにも使用できます。これは、システム値 QUPSMGQ に指定されているメッセージ待ち行列を割り振らない、無停電電源装置プログラムの 2 番目のバージョンを作成することによって行えます。(つまり、ALCOBJ コマンドは使用しません。) 制限状態にあるときにこのプログラムを使用するには、特定の機能 (SAVSYS など) を使用する前に、次のコマンドを入力します。

```
CHGMSGQ MSGQ(LIB/MSGQ) DLVRY(*BREAK)
PGM(LIB/PGM)
```

(LIB/MSGQ) はシステム値 QUPSMGQ で指定されているメッセージ待ち行列の名前であり、(LIB/PGM) は修正した無停電電源装置処理プログラムの名前です。これで、電源障害が生じると、SAVSYS などの機能が実行中であっても、電源障害メッセージが中断処理プログラムによって処理されるようになります。中断処理プログラムを非活動化するには、ユーザーがサインオフするか、または次のように入力してください。

```
CHGMSGQ MSGQ(LIB/MSGQ) DLVRY(*HOLD)
PGM(*DSPMSG)
```

中断処理プログラムを非活動化したら、すぐにサブシステムと通常の無停電電源装置処理プログラムを開始してください。

## 例

11 ページの『例: 電源管理 CL プログラム』

14 ページの『例: 電源管理 CL プログラムのテスト』



---

## 第 5 章 例: 電源管理プログラム

特定のシステム要件に応じて、サンプル・プログラムを変更できます。また、ユーザーが作成した電源管理プログラムをテストすることもできます。

- 『例: 電源管理 CL プログラム』
- 14 ページの『例: 電源管理 CL プログラムのテスト』

注: 重要な法的情報については、3 ページの『第 1 章 コードに関する特記事項』をお読みください。

---

### 例: 電源管理 CL プログラム

サンプル・プログラムを完全無停電電源装置と一緒に使用できます。このサンプル CL プログラムはこの状態のままでも正しく機能しますが、特定のシステム要件に合うように調整が必要です。たとえば、システムに特有のエラー条件を監視することによって、付加的な回復方法をプログラムに追加できます。さらに、システムの通常遮断に必要なステップを実行する、ユーザー作成プログラムを提供する必要もあります。これらのステップには、ジョブ待ち行列の保留、メッセージの送信、およびサブシステムの終了が含まれます。プログラムは、システムが電源遮断する前に電源異常が終了した場合には、通常の操作を再始動する必要があります。

注: 重要な法的情報については、3 ページの『第 1 章 コードに関する特記事項』をお読みください。

プログラムは次のことを行います。

1. 電源管理プログラムはシステム値 QUPSMMSGQ を変数 &LIB および &MSGQ から検索します。これは必ずしも必要ではありませんが、プログラムが始動するたびに正しいメッセージ待ち行列が割り振られるようにする上で役立ちます。次に、プログラムはメッセージ待ち行列を除去し (すでに存在する場合)、再び作成します。このステップにより、メッセージ待ち行列の消去や、メッセージ待ち行列に損傷がある場合に生じる問題を避けることができます。
2. メッセージ待ち行列が作成されたら、プログラムはメッセージ待ち行列を排他的に割り振らなければなりません (ALCOBJ コマンド)。

注: システム値 QUPSDLYTIM を \*NOMAX に設定した場合は、次のいずれかの方法を使用して、システム値 QUPSMMSGQ に指定されているメッセージ待ち行列を割り振ってください。

- コマンド CHGMSGQ MSGQ(UPSLIB/UPSMMSGQ) MODE(\*BREAK) を使用する。
- 電源管理プログラムに ALCOBJ コマンドを含める。

上記の方法のうちのどちらか 1 つだけを使用してください。

ユーザーまたはプログラムがメッセージ待ち行列を割り振っていない状態で電源異常が生じた場合、システムは即時に電源遮断を行います。

3. 電源管理プログラム例のラベル A では、メッセージの受信 (RCVMSG) コマンドが発行され、メッセージ待ち行列に送信されたメッセージを判別します。RCVMSG コマンドは、プログラム全体の待ち時間 (WAIT パラメーター) を判別するときにも使用します。

電源管理プログラム例の行 25.00 で指定されている、RCVMSG コマンドの WAIT パラメーターの値により、プログラムは 600 秒 (10 分) 待機することになります。10 分経過すると、プログラムはジョブに対して制御終了が行われたかどうかを確認します (ENDSBS または ENDJOB コマンドを使用)。こうして、非終了プログラムが ENDJOB または ENDSBS コマンドを遅らせることがないようにします。

ENDSBS \*IMMED または ENDJOB \*IMMED を使用する場合、プログラムのこの部分は除去できません。RCVMSG コマンドの WAIT パラメーターの値を \*MAX に変更できます。RCVMSG コマンドに指定されているメッセージ待ち行列にシステムがメッセージを送信する場合、RCVMSG コマンドが直ちに実行されます。これは、WAIT パラメーターに指定されている値に関係なく行われます。

4. RCVMSG コマンドによって受信されたメッセージが CPF1816 (システムの通常電源障害) である場合、プログラムは、これが単なる短時間の電源障害であるかどうかを確認します。プログラムは、WAIT パラメーターに 10 秒が指定されている、2 番目の RCVMSG コマンドを実行します (サイトに適した秒数を決定してください)。

指定した 10 秒以内に RCVMSG コマンドによって受信されたメッセージが CPF1817 (システムの通常電源の回復) である場合、電力は回復しています。プログラムはラベル A に戻り、循環を再び開始します。

制限の 10 秒に達してもメッセージが受信されなかった場合、電源障害は 10 秒を超えたため、付加的なステップが必要になります。ここで、次の処理を行うユーザー作成プログラムを呼び出すことができます。

- HLDJOBQ コマンドを使用し、長時間実行されている特定のバッチ・ジョブを保留にする
  - 影響を受けていないリモート・ユーザーに通知を行う
  - ジョブとサブシステムを正常な方法で終了させる
5. 電源管理プログラム例のラベル B では、プログラムは電源障害を乗り切ろうとします。プログラムは現在の時間を検索し、この情報を &START という名前の CL 変数に入れます。現在の時間は、無停電電源装置の時間がどれだけ残っているかを判別するために使用されます。

3 番目の RCVMSG コマンドが実行され、&WAIT (プログラムの前の部分で変更された) という CL 変数が WAIT パラメーターの値を決定します。CL 変数 &WAIT は、無停電電源装置が提供可能な予備電力の量です。ラベル A の &WAIT 変数の値は、無停電電源装置が提供可能な予備電力の量に調整されなければなりません。(予備電力の量は秒単位で計測されます。)

プログラム例では、&WAIT 変数の値は 1200 秒 (20 分) に設定されています。その時間内にメッセージ CPF1817 (システムの通常電源の回復) が受信されると、電力が復元されたということになり、他のプログラムを呼び出して通常のシステム操作を再始動できるようになります。ここで、プログラムはラベル A に戻り、循環を再び開始します。1200 秒経過してもメッセージ CPF1817 が送信されない場合、RCVMSG はブランクのメッセージ ID (CPF1817 とは異なる) を返します。これは、電力が復元されておらず、直ちにシステム電源遮断が開始されることを示しています。1200 秒の間に CPF1817 以外のメッセージが受信される場合、次の処置がとられます。

- a. プログラムは、現在の時間を検索し、1200 秒の待ち時間のうちどれだけ経過したかを計算します。
- b. プログラムはその差を減算し、その値を反映するように CL 変数 &WAIT を変更します。
- c. プログラムはラベル B に戻り、無停電電源装置が提供する残りの電力を使用します。

プログラムのこの部分では、日付が変更されるかどうかを確認します。これは、電源異常が異なる日に生じた場合に必要です。

```

SEQNBR*...+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 ...
1.00      PGM
2.00      DCL      VAR(&UPSMGQ)  TYPE(*CHAR)  LEN(20)
3.00      DCL      VAR(&LIB)     TYPE(*CHAR)  LEN(20)
4.00      DCL      VAR(&MSGQ)    TYPE(*CHAR)  LEN(20)
5.00      DCL      VAR(&MSGID)   TYPE(*CHAR)  LEN(7)
6.00      DCL      VAR(&ENDSTS)  TYPE(*CHAR)  LEN(1)
7.00      DCL      VAR(&WAIT)    TYPE(*DEC)   LEN(6)
8.00      DCL      VAR(&HOUR)    TYPE(*DEC)   LEN(6)
9.00      DCL      VAR(&MIN)     TYPE(*DEC)   LEN(6)
10.00     DCL      VAR(&SEC)     TYPE(*DEC)   LEN(6)
11.00     DCL      VAR(&TIME)    TYPE(*CHAR)  LEN(6)
12.00     DCL      VAR(&START)   TYPE(*DEC)   LEN(6)
13.00     DCL      VAR(&END)     TYPE(*DEC)   LEN(6)
14.00     DCL      VAR(&RESULT)  TYPE(*DEC)   LEN(6)
15.00
16.00     RTVSYSVAL  SYSVAL(QUPSMGQ) RTNVAR(&UPSMGQ)
17.00     CHGVAR    VAR(&MSGQ)  VALUE(%SST(&UPSMGQ 1 10))
18.00     CHGVAR    VAR(&LIB)   VALUE(%SST(&UPSMGQ 11 10))
19.00     DLTMSGQ    MSGQ(&LIB/&MSGQ)
20.00     MONMSG    MSGID(CPF2105) /* Message queue not found. */
21.00     CRTMSGQ    MSGQ(&LIB/&MSGQ) TEXT('UPS Power handling +
22.00                                     program message queue') AUT(*EXCLUDE)
23.00     ALCOBJ    OBJ((&LIB/&MSGQ *MSGQ *EXCL))
24.00
25.00     A:  RCVMSG    MSGQ(&LIB/&MSGQ) WAIT(600) RMV(*YES) +
26.00                                     MSGID(&MSGID)
27.00     IF          COND(&MSGID *NE CPF1816) THEN(DO)
28.00     RTVJOB    ENDSTS(&ENDSTS)
29.00     IF          COND(&ENDSTS *EQ '1') THEN(GOTO CMDLBL(ENDPGM))
30.00     GOTO      CMDLBL(A)
31.00     ENDDO
32.00
33.00     /* Check to see if this is a short power outage. */
34.00     IF          COND(&MSGID *EQ CPF1816) THEN(DO)
35.00     RCVMSG    MSGQ(&LIB/&MSGQ); WAIT(10) RMV(*YES) +
36.00                                     MSGID(&MSGID); /* Wait ten seconds)
37.00     IF          COND(&MSGID *EQ CPF1817) THEN(GOTO CMDLBL(A))
38.00     ENDDO
39.00
40.00     /* Power outage was longer than 10 seconds. */
41.00     CALL      PGM(LIB/PGM) /* User program that prepares +
42.00                                     system for possible shutdown. */
43.00

```

```

44.00      /* Check to see if this is a long power outage. */
45.00      CHGVAR      VAR(&WAIT); VALUE(01200) /* 20 minutes. */
46.00 B:  RTVSYSVAL  SYSVAL(QTIME) RTNVAR(&TIME);
47.00      CHGVAR      VAR(&HOUR); VALUE(%SST(&TIME 1 2))
48.00      CHGVAR      VAR(&MIN); VALUE(%SST(&TIME 3 2))
49.00      CHGVAR      VAR(&SEC); VALUE(%SST(&TIME 5 2))
50.00      CHGVAR      VAR(&START); VALUE((&SEC); + (&MIN * 60) + +
51.00              (&HOUR * 3600))
52.00      RCVMSG     MSGQ(&LIB/&MSGQ); WAIT(&WAIT); RMV(*YES) +
53.00              MSGID(&MSGID);
54.00      IF          COND(&MSGID *EQ CPF1817) THEN(DO)
55.00      CALL        PGM(LIB/PGM) /* User program that restarts +
56.00              system operations. */
57.00      GOTO        CMDLBL(A)
58.00      ENDDO
59.00
60.00      IF          COND(&MSGID *NE CPF1817) THEN(DO)
61.00      RTVSYSVAL  SYSVAL(QTIME) RTNVAR(&TIME)
62.00      CHGVAR      VAR(&HOUR) VALUE(%SST(&TIME 1 2))
63.00      CHGVAR      VAR(&MIN) VALUE(%SST(&TIME 3 2))
64.00      CHGVAR      VAR(&SEC) VALUE(%SST(&TIME 5 2))
65.00      CHGVAR      VAR(&END) VALUE((&SEC) + (&MIN * 60) + +
66.00              (&HOUR * 3600))
67.00      CHGVAR      VAR(&RESULT); VALUE(&END - &START)
68.00      IF          COND(&RESULT < 0) THEN(CHGVAR VAR(&RESULT) +
69.00              VALUE(86400 + &RESULT)) /* Check for +
70.00              change of day. 86400 = 24 hours. */
71.00      IF          COND(&RESULT *GE &WAIT) THEN(PWRDWSYS +
72.00              OPTION(*IMMED) /* UPS battery reserve has +
73.00              expired. */
74.00      CHGVAR      VAR(&WAIT) VALUE(&WAIT - &RESULT) /* UPS +
75.00              battery reserve has not expired. */
76.00      GOTO        CMDLBL(B)
77.00      ENDDO
78.00
79.00
80.00      ENDPGM:    DLCOBJ      OBJ((&LIB/&MSGQ *MSGQ *EXCL))
81.00      ENDPGM

```

## 関連する例

『例: 電源管理 CL プログラムのテスト』

---

## 例: 電源管理 CL プログラムのテスト

電源管理プログラムを作成したら、プログラム・メッセージの送信 (SNDPGMMSG) コマンドとジョブの遅延 (DLYJOB) コマンドを使った簡単な CL プログラムを作成して、その電源管理プログラムをテストできます。DLYJOB コマンドの DLY パラメーター値は、テストの必要にかなう値に設定してください。

### 注:

1. 重要な法的情報については、3 ページの『第 1 章 コードに関する特記事項』をお読みください。
2. 無停電電源装置プログラムをテストするときに、PWRDWSYS、ENDJOB、および ENDSBS などのコマンドは、コマンドが実行されていることを示す SNDMSG コマンドに置き換えてください。

```

SEQNBR*...+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 ...+... 7....
1.00      PGM
2.00      DLYJOB      DLY(120) /* Wait for 2 minutes. */
3.00      SNDPGMMMSG  MSGID(CPF1816) MSGF(QCPFMSG) +
4.00                        TOMSGQ(UPSLIB/UPSMMSGQ) /* Power failure
5.00                        message. */
6.00      DLYJOB      DLY(5) /* Wait for 5 seconds. */
7.00      SNDPGMMMSG  MSGID(CPF1817) MSGF(QCPFMSG) +
8.00                        TOMSGQ(UPSLIB/UPSMMSGQ) /* Power restored
9.00                        message. */
10.00     ENDPGM

```

図1. 電源管理プログラム例のテスト

### 関連する例

11 ページの『例: 電源管理 CL プログラム』



---

## 第 6 章 制御遮断の概念

制御遮断の概念と定義について調べるには、ここから開始してください。

- 『バッテリー・バックアップ装置』
- 『電源管理プログラムが存在しない場合の無停電電源装置条件の処理』
- 『無停電電源装置の IPL 考慮事項』
- 18 ページの『停電時の制御遮断』
- 18 ページの『電源回復 IPL システム値 (QPWRRSTIPL)』
- 19 ページの『無停電電源装置』
- 19 ページの『無停電電源装置遅延時間システム値 (QUPSDLYTIM)』
- 22 ページの『無停電電源装置メッセージ待ち行列システム値 (QUPSMMSGQ)』
- 22 ページの『無停電電源装置メッセージ』
- 24 ページの『無停電電源装置からのバッテリー低下状態信号』

---

### バッテリー・バックアップ装置

バッテリー・バックアップ装置 (BBU) は、無停電電源装置がない場合、あるいは無停電電源装置が故障している場合の保護策です。BBU は 30 秒の実行時間の猶予を与えます。AC 電源が 30 秒以内に回復しない場合、システムは直ちに制御遮断を行います。

システム 620、640、650、720、730、740、830、SB1、SB2、SB3 には、BBU が装備されています。

---

### 電源管理プログラムが存在しない場合の無停電電源装置条件の処理

電源管理プログラムがないことを指定するには、QUPSMMSGQ のデフォルトである QSYSOPR を使用します。このデフォルトを使用すると、システムは電源に関連したすべてのメッセージを QSYSOPR に送信します。QUPSDLYTIM を \*NOMAX 以外の値に設定する必要があります。

通常、ワークステーション装置に対する電力の供給は行われません。通常電源が中断すると、システムは活動状態を保ちますが、ワークステーションはほとんどの場合異常終了します。システムが無停電電源装置を利用して動作している間に通常電源が回復すると、システムは活動状態を保ち、ワークステーション・ジョブは再始動可能になります。

QUPSDLYTIM タイマーが終了するか、またはバッテリー低下信号が出されると、システムは主記憶域を保管し、電源遮断を行います。無停電電源装置とシステム・サイズに合った QUPSDLYTIM の値を選択してください。

無停電電源装置の電力供給中にシステムの電源を遮断する場合、通常電源の回復時に IPL を実行するかどうかは QPWRRSTIPL によって判別されます。デフォルトでは、IPL を実行しません。

---

### 無停電電源装置の IPL 考慮事項

システムが IPL を実行するときに、ライセンス内部コードはさまざまな内部スイッチを調べ、システムが正しく電源遮断を行ったかどうかを確認します。システムの電源遮断 (PWRDWN SYS) コマンドが正常終了した場合にのみ、iSeries サーバーは正しく電源遮断を行います。他の方法で遮断した場合、OS/400 プログラムは次回の IPL を異常と判断します。ライセンス内部コードは、システムが主記憶域を保管し、電源



遮断手順を正常に行った場合に、IPL が正常であると判断します。どちらの電源遮断手法も正常に行われなかった場合、ライセンス内部コードは、次の IPL で種々の回復機能を実行します。

異常な IPL が生じると、OS/400 プログラムは追加の回復機能を実行します。在席 IPL では、これらの機能のいくつかを制御できます。不在 IPL (『電源回復 IPL システム値 (QPWRRSTIPL)』または時間指定 IPL によって生じる) の場合、システムは現在設定されている処置を行います。

電力がバッテリー・バックアップ装置 (BBU) によって、または無停電電源装置によって提供されているときに、システムの電源遮断 (PWRDWN SYS) コマンドが実行されると、システムは次の IPL までジョブ・ログの書き出しをすべて保留します。システムがこのような方法で PWRDWN SYS を実行するのは、処理の量を最小限にとどめるためです。BBU の動作中にシステムが IPL を実行することはありません。

通常電源がオフで、システムが無停電電源装置を利用して動作している場合、システムは IPL を実行できません。これは時間指定 IPL またはリモート IPL には当てはまりません。通常電源に障害がある場合は、手動でしか IPL を開始できません。

無停電電源装置メッセージの詳細については、22 ページの『無停電電源装置メッセージ』を参照してください。

---

## 停電時の制御遮断

停電時の制御遮断機構は、通常電源の停電後に、システムを正常な方法で電源遮断することを可能にします。電源ロス時の制御遮断機構は、iSeries サーバーに無停電電源装置を接続した場合にのみ利用できます。

通常電源が QUPSDLYTIM システム値によって指定された時間内に回復しない場合、ライセンス内部コードが各ジョブに対し、次の命令境界で終了するという信号を送ります。通常、ジョブは次の命令境界にあるか、または間もなくそこに達することになります。しかし、実行に長時間を要する命令 (たとえば、アクセス・パスを構築したり、プログラムを作成したりする命令) は、許可されている時間内に完了しないかもしれません。命令境界に達するための固定された内部時間が経過すると、主記憶域の変更されたページが補助記憶装置に書き込まれ、システムは異常遮断し、次いで電源がオフになります。

停電時の制御遮断が正常に終了すると、ライセンス内部コードは次のシステム・プログラムの初期プログラム・ロード (IPL) を正常なものと見なしますが、オペレーティング・システムはこれを正常なものとは見なしません。停電時の制御遮断が完了するまで無停電電源装置のバッテリーがもたない場合、ライセンス内部コードは次の IPL を異常なものとして見なします。

QUPSDLYTIM システム値の詳細については、19 ページの『無停電電源装置遅延時間システム値 (QUPSDLYTIM)』を参照してください。

---

## 電源回復 IPL システム値 (QPWRRSTIPL)

この値は、通常電源に障害が起こってシステムが終了する時、およびその後で障害が回復する時に起こる事柄を制御します。デフォルトは 0 (使用不可能) です。これは、通常電源の回復時にシステムが IPL を実行しないことを指定します。

次の条件のいずれかが当てはまる場合は、通常、この値を 0 のままにしておきます。

- システムを再び手動で実行する場合
- もう一度 IPL を実行するのに十分なほどバッテリーが再充電されているかどうかを確認する電源管理プログラムがある場合



---

## 無停電電源装置

無停電電源装置は、通常電源に障害が生じたときに AC 電源を供給します。多くの場合、無停電電源装置のバックアップ時間は制限されています。

無停電電源装置は、通常電源が途絶えている間、 iSeries サーバーおよび関連するすべての DASD コントローラーおよびデバイスに、電力を供給します。無停電電源装置の稼働時間は、停電時の制御遮断を行うのに十分な時間に設定する必要があります。

3 つのシステム値が無停電電源装置に影響します。これらのシステム値は、無停電電源装置が接続された時に電力供給信号が変わるのに応じてシステムがとる処置を定義します。

QUPSMMSGO  
QUPSDLTYTIM  
QPSRRSTIPL

---

## 無停電電源装置遅延時間システム値 (QUPSDLTYTIM)

QUPSDLTYTIM は、『無停電電源装置』遅延タイマーのシステム値です。QUPSDLTYTIM の値は、主記憶域を保管し、システムの電源遮断を行うまでシステムが待機する時間を制御します。待機時間が終了する前に通常電源が回復した場合、システムはタイマーをリセットします。待機時間を超えると、システムは主記憶域の保管を開始するか、または制御遮断を行います。

QUPSDLTYTIM 値として次の 3 つの選択肢があります。

- \*BASIC または \*CALC
- 数値
- \*NOMAX


QUPSDLTYTIM システムによって指定される時間としては、次のようなものがあります。

- QUPSDLTYTIM が \*NOMAX に設定されている場合は無限 (電源管理プログラムを持っていない限りません)。
- QUPSDLTYTIM が \*BASIC または \*CALC に設定されている場合は固定内部遅延。
- QUPSDLTYTIM によって指定される代替値。値は正数で、秒単位でなければなりません (ゼロは遅延なしを意味します)。

QUPSDLTYTIM が \*NOMAX に設定されている場合は、以下の条件を満たしていなければなりません。そうでない場合、システムは即時電源遮断を開始します。

- QUPSMMSGQ システム値で指定されているメッセージ・キューが存在しなければならない。
- メッセージ・キューがワークステーション・メッセージ・キュー (または QSYSOPR) である場合、中断または通知モードでなければならない。
- メッセージ・キューがワークステーション・メッセージ・キューでない場合、それらはジョブによって割り振られなければならない。

これらの選択肢とその意味を図示します。



RV4N350-02

図2. QUPSDLYTIM 値と処置

QUPSDLYTIM のデフォルト値は \*CALC です。QUPSDLYTIM を \*CALC に設定したままにしておくと、無停電電源装置の効果を無効にすることになります。\*BASIC と \*CALC は、OS/400 の V3R6 またはそれ以降のリリースのシステムでは、同じ働きをします。

QUPSDLYTIM が \*BASIC または \*CALC に設定されている場合は、システムは固定された間隔遅延 (200 秒) の後に制御遮断を行います。無停電電源装置を持っている場合には、数値を指定することができます。『QUPSDLYTIM 値の決定』を参照してください。

### QUPSDLYTIM 値の決定

QUPSDLYTIM の値を決定するには、次のことを知っておく必要があります。

- 所定の負荷をかけた場合の無停電電源装置の停止時間 (バッテリーの評価時間)
- システムが主記憶域の保管と電源遮断に要する時間

#### 使用可能なバッテリー実行時間

使用可能なバッテリー実行時間は、一種の動的な変数です。以下の質問を考慮して、システムの使用可能なバッテリーを評価してください。

- 電力は不安定か?

- バッテリーは以前に放電されたか?
- バッテリーは完全に充電されているか?

実際のバッテリー実行時間は、容量に依存します。バッテリーが完全に充電されている場合でも、100% の容量が利用できるとは限りません。多くの場合、バッテリーの見積容量の 20% ~ 50% は、4 年から 5 年で失われます。これは、室温によっても異なります。動作温度が高くなると、容量の無駄が多くなる傾向があります。実際のバッテリー実行時間は、放電の負荷にも関係します。UPS が供給する負荷が高くなると、それだけ持続時間は短くなります。システムのバッテリーが特定の充電レベルを下回ると、UPS はバッテリー低下状態信号を出します。24 ページの『無停電電源装置からのバッテリー低下状態信号』は、遮断メカニズムに影響を与えます。

主記憶域の保管および電源遮断に要する時間は、厳密なものではありません。保管時間は、ディスクに書き込まれていない主記憶域内の変更数によって異なります。また、使用可能なディスク・アームの数も関係があります。ディスク・アームが多ければ多いほど、システムはより速く主記憶域をディスクに書き込むことができます。さらに、システムの電源遮断に要する時間は、ジョブの数と、それらを終了するのに必要な平均時間にも依存します。通常、ジョブは命令境界に近くなりますが、命令によっては実行に長時間を要するものもあります。次に挙げるのは、システムの遮断に要する時間 (分) を、最悪の事態を想定して見積もるための公式です。

$$((0.0554 * (\text{MS サイズ (MB 単位)})) / (\# \text{ ディスク・アーム数})) + 1.6 = \# \text{ 時間 (分)}$$

特定のシステム上のディスクに主記憶域を書き込むのに要する時間の見積もりについては、以下の表を参照してください。多くの場合、システムの遮断に要する時間は、この値よりかなり少なくなります。

表 1. ディスクに主記憶域を書き込むために要する時間 (分単位)

アーム	32MB	64MB	128MB	256MB	512MB	1GB	2GB	4GB	8GB	16GB	32GB	64GB	128GB
2	2.5	3.4	5.1	8.7	15.8	30.0	58.4	115.1	228.7				
4	2.0	2.5	3.4	5.1	8.7	15.8	30.0	58.4	115.1	228.7			
8	1.8	2.0	2.5	3.4	5.1	8.7	15.8	30.0	58.4	115.1	228.7		
16	1.7	1.8	2.0	2.5	3.4	5.1	8.7	15.8	30.0	58.4	115.1	228.7	
32	1.6	1.7	1.8	2.0	2.5	3.4	5.1	8.7	15.8	30.0	58.4	115.1	228.7
64		1.6	1.7	1.8	2.0	2.5	3.4	5.1	8.7	15.8	30.0	58.4	115.1
128			1.6	1.7	1.8	2.0	2.5	3.4	5.1	8.7	15.8	30.0	58.4
256				1.6	1.7	1.8	2.0	2.5	3.4	5.1	8.7	15.8	30.0
512					1.6	1.7	1.8	2.0	2.5	3.4	5.1	8.7	15.8
1024						1.6	1.7	1.8	2.0	2.5	3.4	5.1	8.7

主記憶域内のすべてのページが変更されており、それらの変更がまだディスクに書き込まれていないことを前提とします。ディスクに書き込む必要のあるページがほとんどない場合には、それに比例して、書き込みに要する時間も少なくなります。

## QUPSDLYTIM 値およびアクション・フローチャートに関する解説

フローチャートでは、QUPSDLYTIM システム値の値およびアクションを示しています。フローチャートは、チャートの最上部にある 3 つのボックスから始まります。

最初のボックスは、「ユーティリティ障害、および UPS がない、または UPS 障害」というラベルが付いています。このボックスからの矢印は、「30 秒の遅延」というラベルが付いている別のボックスに進みます。さらに別の矢印は、「30 秒の遅延」ボックスから「CPM」というラベルが付いているボックスに進みます。

2 番目のボックスは、「ユーティリティ障害、および UPS は正しく作動している」というラベルが付いています。このボックスからの矢印は「QUPSDLYTIM SETTING」というラベルが付いている別のボックスに進みます。このボックスは、「QUPSDLYTIM SETTING」用の 3 つの設定 (それぞれ「\*NOMAX」、「数値」、および「\*CALC または \*BASIC」) を示しています。

「\*NOMAX」が選択される場合、矢印は「戻る」というラベルが付いている別のボックスに進みます。「数値」が選択される場合、「QUPSDLYTIM SETTING」ボックスから出ている矢印は、「ユーザー定義の遅延」というラベルが付いている別のボックスに進みます。「ユーザー定義の遅延」ボックスからの別の矢印は「遮断」というラベルが付いているボックスに進みます。「\*CALC または \*BASIC」が選択される場合、「QUPSDLYTIM SETTING」ボックスから出ている矢印は、「45 秒の遅延」というラベルが付いている別のボックスに進みます。「45 秒の遅延」ボックスは、「AVAILABLE FEATURES」というラベルが付いている別のボックスに進みます。「AVAILABLE FEATURES」ボックスには 2 つの設定があります。これは「CPM」と「CPM がないが UPS は使用可能」という設定です。「CPM」が指定される場合、矢印は「AVAILABLE FEATURES」ボックスから「CPM」というラベルが付いているボックスに進みます。「CPM がないが UPS は使用可能」が指定される場合、矢印は「遮断」というラベルが付いているボックスに進みます。

3 番目のボックスは、「ユーティリティ障害および UPS バッテリー低下」というラベルが付いています。このボックスは、「AVAILABLE FEATURES」というラベルが付いているボックスに進みます。「CPM」が指定される場合、矢印は「AVAILABLE FEATURES」ボックスから「CPM」というラベルが付いているボックスに進みます。「CPM がないが UPS は使用可能」が指定される場合、矢印は「遮断」というラベルが付いているボックスに進みます。

## 無停電電源装置メッセージ待ち行列システム値 (QUPSMMSGQ)

無停電電源装置メッセージ待ち行列 (QUPSMMSGQ) システム値は、電源装置メッセージの送信先メッセージ待ち行列を決定します。

システムは、システム値で指定されている値にかかわらず、電源機構に関するメッセージをシステム・オペレーター (QSYSOPR) メッセージ待ち行列に送信します。異なるメッセージ待ち行列を指定すると、そのメッセージ待ち行列も同じ電源機構メッセージを受信することになります。次のいずれかのものがある場合は、異なるメッセージ待ち行列を指定してください。

- 電源機構メッセージを受信するための別のメッセージ待ち行列 (たとえば、データ処理マネージャーのメッセージ待ち行列)
- 無停電電源装置と関連するイベントを扱うプログラム

## 無停電電源装置メッセージ

メッセージ ID	メッセージ・テキスト	追加情報
<b>CPF1816</b>	&1 にシステムの通常電源の障害が起こった。	
<b>CPF1817</b>	&1 にシステムの通常電源が復元された。	システムの電源が通常電源に切り替えられました。

メッセージ ID	メッセージ・テキスト	追加情報
<b>CPI1819</b>	<p>システム終了中。電源障害メッセージは監視されていません。システムは理由 &amp;3 で終了します。</p> <p>注 1: ライブラリー &amp;2 の QUPSMGQ システム値で指定されたメッセージ待ち行列 &amp;1 がユーザー・プログラムまたはワークステーションに割り振られていない。</p> <p>注 2: ライブラリー &amp;2 の QUPSMGQ で指定されたメッセージ待ち行列 &amp;1 はワークステーションに割り振られているか、あるいはシステム操作員メッセージ待ち行列 (ライブラリー QSYS の QSYSOPR) であるが、このメッセージ待ち行列が *BREAK または *NOTIFY モードになっていない。</p>	
<b>CPI0961</b>	無停電電源装置 (UPS) はもう接続されていない。	
<b>CPI0962</b>	現在無停電電源装置 (UPS) が接続されている。	
<b>CPI0963</b>	システムは補助電源になっている。	システムは現在補助電源で実行中です。
<b>CPI0964</b>	バッテリー低下状態になっている。	外部無停電電源装置 (UPS) または内部バッテリーが、バッテリーの低下状態を示しています。この状態で通常電源が切れた場合には、システムは即時遮断を開始する場合があります。詳細については、無停電電源装置の解説書を参照してください。
<b>CPI0965</b>	システム装置のバッテリー・バックアップ機構に故障がある。	システム装置のバッテリーの故障、またはバッテリー・バックアップ機構のバッテリー・チャージャーの故障が考えられます。弊社技術員に連絡してください。
<b>CPI0966</b>	拡張装置のバッテリー・バックアップ機構に故障がある。	拡張装置のバッテリーの故障、またはバッテリー・バックアップ機構のバッテリー・チャージャーの故障が考えられます。弊社技術員に連絡してください。
<b>CPI0973</b>	バッテリー低下状態はもう存在していない。	外部無停電電源装置または内部バッテリーのバッテリー低下状態はもう存在していません。UPS の資料を参照してください。
<b>CPI0974</b>	UPS がバイパスされた。	通常電源の障害が起こった場合には、無停電電源装置がシステム電源を供給できません。システムが異常終了することになります。
<b>CPI0975</b>	もう UPS はバイパスされていない。	無停電電源装置 (UPS) は、もうバイパスされていません。

メッセージ ID	メッセージ・テキスト	追加情報
<b>CPI0976</b>	メッセージ &1 の通知が正常に行われなかった。	QUPSMGQ システム値に指定されたライブラリー &3 のメッセージ待ち行列 &2 に、&1 メッセージを送ることができません。
<b>CPI0981</b>	自動 IPL できなかった。	次のいずれかの理由のために、システム値 QPWRSTIPL によって指定された通常電源回復後の自動 IPL が行われていません。 <ul style="list-style-type: none"> <li>• IPL 中に通常電源の障害があり、バッテリー低下状態が検出されている。</li> <li>• IPL 中に通常電源の障害があり、IPL の完了前に、システム値 QUPSDLYTIM に指定された無停電電源装置遅延時間が超えた。</li> </ul>
<b>CPI0994</b>	システムの電源が復元された。	&1 の時点で、システムの電源が主電源に切り替えられました。 &2 秒間通常電源に障害がありました。この間、システムはどのアプリケーションの処理も実行していません。通常電源の障害が続く場合には、システムの電源を遮断してください (PWRDWN SYS コマンド)。

## 無停電電源装置からのバッテリー低下状態信号

システムのバッテリーが特定の充電レベルを下回ると、19 ページの『無停電電源装置』はバッテリー低下状態信号を出します。次の条件が当てはまる場合、無停電電源装置からのバッテリー低下信号は、システムが停電時の制御遮断を実行できるようにします。

- 『システムの計画』のトピックに説明されているように、iSeries サーバーと UPS との間で 4 本のワイヤー通信を使用することを選択している場合
- ユーティリティ障害信号が活動状態にある場合

制御遮断が即時に行われます。システムが通常電源を利用して稼働しており、無停電電源装置がバッテリー低下信号を送っている場合、システムは稼働したままで CPI0964 メッセージを送信します。このメッセージの詳細については、22 ページの『無停電電源装置メッセージ』を参照してください。しかし、通常電源が失われると、システムはこの条件のもとで直ちに遮断メカニズムを開始します。

多くの UPS は、実行時間の残りが 2 分程になったところでバッテリー低下信号を送信するよう、工場ですべて事前設定されています。UPS の種類によっては、この時間を調整できます。システムが停電時の制御遮断を実行するのに必要とする時間に合わせてこの時間を設定するのが理想と言えます。工場での UPS の事前設定が、システムの通常の遮断に十分であると考えないでください。停電時の制御遮断を実行するのに必要な時間を計算し、QUPSDLYTIM に適切な値を使用し、UPS のバッテリー低下信号が出される時間を調整してください (調整可能な場合)。

以下の図では、QUPSDLYTIM 機能の進行を時間の経過にそって左から右に示しています。






図3. QUPSDLYTIM 機能の時間軸

## QUPSDLYTIM 機能の時間軸に関する図の解説

時間軸は、停電時からシステム遮断までに何が起きるかを示しています。この時間は、バッテリーの見積寿命と同じです。時間軸は、通常電源がオフになる時点で始まります。同時に、UPS またはバッテリー電源装置が活動状態になります。この時点で、ユーザー・プログラム機能 (ある場合) および QUPSDLYTIM 値のタイムアウトが有効になります。その時間が満了すると、主記憶域が保管され、電源が遮断されます。その後も、安全を考慮して、必ずいくらかのバッテリー寿命が残されます。









Printed in Japan